

## Rules for integrands of the form $(a + b \tan[c + d x^n])^p$

1:  $\int (a + b \tan[c + d x^n])^p dx$  when  $\frac{1}{n} \in \mathbb{Z}^+ \wedge p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If  $-1 \leq n \leq 1 \wedge n \neq 0$ , then  $F[x^n] = \frac{1}{n} \text{Subst}[x^{\frac{1}{n}-1} F[x], x, x^n] \partial_x x^n$

Note: If  $\frac{1}{n} \in \mathbb{Z}^-$ , resulting integrand is not integrable.

Rule: If  $\frac{1}{n} \in \mathbb{Z}^+ \wedge p \in \mathbb{Z}$ , then

$$\int (a + b \tan[c + d x^n])^p dx \rightarrow \frac{1}{n} \text{Subst}\left[\int x^{\frac{1}{n}-1} (a + b \tan[c + d x])^p dx, x, x^n\right]$$

Program code:

```
Int[(a_.+b_.*Tan[c_.+d_.*x_^n_])^p_.,x_Symbol] :=  
  1/n*Subst[Int[x^(1/n-1)*(a+b*Tan[c+d*x])^p,x],x,x^n] /;  
FreeQ[{a,b,c,d,p},x] && IGtQ[1/n,0] && IntegerQ[p]
```

```
Int[(a_.+b_.*Cot[c_.+d_.*x_^n_])^p_.,x_Symbol] :=  
  1/n*Subst[Int[x^(1/n-1)*(a+b*Cot[c+d*x])^p,x],x,x^n] /;  
FreeQ[{a,b,c,d,p},x] && IGtQ[1/n,0] && IntegerQ[p]
```

**X:**  $\int (a + b \tan[c + d x^n])^p dx$

Rule:

$$\int (a + b \tan[c + d x^n])^p dx \rightarrow \int (a + b \tan[c + d x^n])^p dx$$

Program code:

```
Int[(a_.+b_.*Tan[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Unintegrable[(a+b*Tan[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x]
```

```
Int[(a_.+b_.*Cot[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Unintegrable[(a+b*Cot[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x]
```

**S:**  $\int (a + b \tan[c + d u^n])^p dx$  when  $u = e + f x$

Derivation: Integration by substitution

Rule: If  $u = e + f x$ , then

$$\int (a + b \tan[c + d u^n])^p dx \rightarrow \frac{1}{f} \text{Subst}\left[\int (a + b \tan[c + d x^n])^p dx, x, u\right]$$

Program code:

```
Int[(a_.+b_.*Tan[c_.+d_.*u_^n_])^p_,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*Tan[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```

```
Int[(a_.+b_.*Cot[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*Cot[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && NeQ[u,x]
```

**N:**  $\int (a + b \tan[u])^p dx$  when  $u = c + d x^n$

Derivation: Algebraic normalization

Rule: If  $u = c + d x^n$ , then

$$\int (a + b \tan[u])^p dx \rightarrow \int (a + b \tan[c + d x^n])^p dx$$

Program code:

```
Int[(a_.+b_.*Tan[u_])^p_.,x_Symbol] :=
  Int[(a+b*Tan[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

```
Int[(a_.+b_.*Cot[u_])^p_.,x_Symbol] :=
  Int[(a+b*Cot[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

Rules for integrands of the form  $(e x)^m (a+b \tan [c+d x^n])^p$ 

1.  $\int x^m (a+b \tan [c+d x^n])^p dx$

1:  $\int x^m (a+b \tan [c+d x^n])^p dx$  when  $\frac{m+1}{n} \in \mathbb{Z}^+ \wedge p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If  $\frac{m+1}{n} \in \mathbb{Z}$ , then  $x^m F[x^n] = \frac{1}{n} \text{Subst}[x^{\frac{m+1}{n}-1} F[x], x, x^n] \partial_x x^n$ Note: If  $\frac{m+1}{n} \in \mathbb{Z}^-$ , resulting integrand is not integrable.Rule: If  $\frac{m+1}{n} \in \mathbb{Z}^+ \wedge p \in \mathbb{Z}$ , then

$$\int x^m (a+b \tan [c+d x^n])^p dx \rightarrow \frac{1}{n} \text{Subst}\left[\int x^{\frac{m+1}{n}-1} (a+b \tan [c+d x])^p dx, x, x^n\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*Tan[c_+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Tan[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IGtQ[Simplify[(m+1)/n],0] && IntegerQ[p]
```

```
Int[x_^m_.*(a_+b_.*Cot[c_+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Cot[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IGtQ[Simplify[(m+1)/n],0] && IntegerQ[p]
```

2:  $\int x^m \tan [c+d x^n]^2 dx$

Note: Although this rule reduces the degree of the tangent factor, the resulting integral is not integrable unless  $\frac{m+1}{n} \in \mathbb{Z}^+$ .

Rule:

$$\int x^m \tan [c + d x^n]^2 dx \rightarrow \frac{x^{m-n+1} \tan [c + d x^n]}{d n} - \int x^m dx - \frac{m-n+1}{d n} \int x^{m-n} \tan [c + d x^n] dx$$

### Program code:

```
Int [x_^m_.*Tan [c_+d_.*x_^n_]^2,x_Symbol] :=
  x^(m-n+1)*Tan [c+d*x^n] / (d*n) - Int [x^m,x] - (m-n+1) / (d*n) *Int [x^(m-n)*Tan [c+d*x^n],x] /;
FreeQ[{c,d,m,n},x]
```

```
Int [x_^m_.*Cot [c_+d_.*x_^n_]^2,x_Symbol] :=
  -x^(m-n+1)*Cot [c+d*x^n] / (d*n) - Int [x^m,x] + (m-n+1) / (d*n) *Int [x^(m-n)*Cot [c+d*x^n],x] /;
FreeQ[{c,d,m,n},x]
```

x.  $\int x^m \tan [a + b x^n]^p dx$  when  $\theta < n < m + 1$

1:  $\int x^m \tan [a + b x^n]^p dx$  when  $\theta < n < m + 1 \wedge p > 1$

Note: Although this rule reduces the degree of the tangent factor, the resulting integrals are not integrable unless  $\frac{m+1}{n} \in \mathbb{Z}^+ \wedge p \in \mathbb{Z}$ .

Rule: If  $\theta < n < m + 1 \wedge p > 1$ , then

$$\int x^m \tan [a + b x^n]^p dx \rightarrow \frac{x^{m-n+1} \tan [a + b x^n]^{p-1}}{b n (p-1)} - \frac{m-n+1}{b n (p-1)} \int x^{m-n} \tan [a + b x^n]^{p-1} dx - \int x^m \tan [a + b x^n]^{p-2} dx$$

### Program code:

```
(* Int [x_^m_.*Tan [a_+b_.*x_^n_]^p,x_Symbol] :=
  x^(m-n+1)*Tan [a+b*x^n]^(p-1) / (b*n*(p-1)) -
  (m-n+1) / (b*n*(p-1)) *Int [x^(m-n)*Tan [a+b*x^n]^(p-1),x] -
  Int [x^m*Tan [a+b*x^n]^(p-2),x] /;
FreeQ[{a,b},x] && LtQ[0,n,m+1] && GtQ[p,1] *)
```

```
(* Int[x^m_.*Cot[a_.*x^n_] ^p_,x_Symbol] :=
-x^(m-n+1)*Cot[a+b*x^n]^(p-1)/(b*n*(p-1)) +
(m-n+1)/(b*n*(p-1))*Int[x^(m-n)*Cot[a+b*x^n]^(p-1),x] -
Int[x^m*Cot[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b},x] && LtQ[0,n,m+1] && GtQ[p,1] *)
```

2:  $\int x^m \tan[a + b x^n]^p dx$  when  $\theta < n < m + 1 \wedge p < -1$

Note: Although this rule reduces the degree of the tangent factor, the resulting integrals are not integrable unless  $\frac{m+1}{n} \in \mathbb{Z}^+ \wedge p \in \mathbb{Z}$ .

Rule: If  $\theta < n < m + 1 \wedge p < -1$ , then

$$\int x^m \tan[a + b x^n]^p dx \rightarrow \frac{x^{m-n+1} \tan[a + b x^n]^{p+1}}{b n (p+1)} - \frac{m-n+1}{b n (p+1)} \int x^{m-n} \tan[a + b x^n]^{p+1} dx - \int x^m \tan[a + b x^n]^{p+2} dx$$

Program code:

```
(* Int[x^m_.*Tan[a_.*x^n_] ^p_,x_Symbol] :=
x^(m-n+1)*Tan[a+b*x^n]^(p+1)/(b*n*(p+1)) -
(m-n+1)/(b*n*(p+1))*Int[x^(m-n)*Tan[a+b*x^n]^(p+1),x] -
Int[x^m*Tan[a+b*x^n]^(p+2),x] /;
FreeQ[{a,b},x] && LtQ[0,n,m+1] && LtQ[p,-1] *)
```

```
(* Int[x^m_.*Cot[a_.*x^n_] ^p_,x_Symbol] :=
-x^(m-n+1)*Cot[a+b*x^n]^(p+1)/(b*n*(p+1)) +
(m-n+1)/(b*n*(p+1))*Int[x^(m-n)*Cot[a+b*x^n]^(p+1),x] -
Int[x^m*Cot[a+b*x^n]^(p+2),x] /;
FreeQ[{a,b},x] && LtQ[0,n,m+1] && LtQ[p,-1] *)
```

**X:**  $\int x^m (a + b \tan[c + d x^n])^p dx$

Rule:

$$\int x^m (a + b \tan[c + d x^n])^p dx \rightarrow \int x^m (a + b \tan[c + d x^n])^p dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Tan[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[x^m*(a+b*Tan[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x]
```

```
Int[x_^m_.*(a_.+b_.*Cot[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Unintegrable[x^m*(a+b*Cot[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x]
```

$$2: \int (e x)^m (a + b \tan[c + d x^n])^p dx$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{(e x)^m}{x^m} = \emptyset$$

Rule:

$$\int (e x)^m (a + b \tan[c + d x^n])^p dx \rightarrow \frac{e^{\text{IntPart}[m]} (e x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m (a + b \tan[c + d x^n])^p dx$$

Program code:

```
Int[(e*x_)^m_.*(a_+b_.*Tan[c_+d_.*x_^n_])^p_,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Tan[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

```
Int[(e*x_)^m_.*(a_+b_.*Cot[c_+d_.*x_^n_])^p_,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Cot[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```



**N:**  $\int (e x)^m (a + b \tan[u])^p dx$  when  $u = c + d x^n$

### Derivation: Algebraic normalization

– Rule: If  $u = c + d x^n$ , then

$$\int (e x)^m (a + b \tan[u])^p dx \rightarrow \int (e x)^m (a + b \tan[c + d x^n])^p dx$$

– Program code:

```
Int[(e*x_)^m_.*(a_.+b_.*Tan[u_])^p_,x_Symbol] :=
  Int[(e*x)^m*(a+b*Tan[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

```
Int[(e*x_)^m_.*(a_.+b_.*Cot[u_])^p_,x_Symbol] :=
  Int[(e*x)^m*(a+b*Cot[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

## Rules for integrands of the form $x^m \operatorname{Sec}[a + b x^n]^p \operatorname{Tan}[a + b x^n]$

1:  $\int x^m \operatorname{Sec}[a + b x^n]^p \operatorname{Tan}[a + b x^n] dx$  when  $n \in \mathbb{Z} \wedge m - n \geq 0$

Derivation: Integration by parts

Note: Dummy exponent  $q = 1$  required in program code so InputForm of integrand is recognized.

Rule: If  $n \in \mathbb{Z} \wedge m - n \geq 0$ , then

$$\int x^m \operatorname{Sec}[a + b x^n]^p \operatorname{Tan}[a + b x^n] dx \rightarrow \frac{x^{m-n+1} \operatorname{Sec}[a + b x^n]^p}{b n p} - \frac{m - n + 1}{b n p} \int x^{m-n} \operatorname{Sec}[a + b x^n]^p dx$$

Program code:

```
Int[x^m_*Sec[a_+b_*x^n_]^p_*Tan[a_+b_*x^n_]^q_,x_Symbol] :=
  x^(m-n+1)*Sec[a+b*x^n]^p/(b*n*p) -
  (m-n+1)/(b*n*p)*Int[x^(m-n)*Sec[a+b*x^n]^p,x] /;
FreeQ[{a,b,p},x] && IntegerQ[n] && GeQ[m,n] && EqQ[q,1]
```

```
Int[x^m_*Csc[a_+b_*x^n_]^p_*Cot[a_+b_*x^n_]^q_,x_Symbol] :=
  -x^(m-n+1)*Csc[a+b*x^n]^p/(b*n*p) +
  (m-n+1)/(b*n*p)*Int[x^(m-n)*Csc[a+b*x^n]^p,x] /;
FreeQ[{a,b,p},x] && IntegerQ[n] && GeQ[m,n] && EqQ[q,1]
```